
The fracpole suite is a collection of programs used to generate Spectre subcircuits that
model components that require a fractional pole impedance. The suite includes a program for generating the fractional pole impedance itself, a program for generating models of inductors that include skin effect, and a program that generates models of
capacitors that include dielectric absorption.

Background

1.0 Background
A fracpole is a basic element that can be used to construct models of common components. It exhibits a fractional impedance pole. Such behavior is useful for modeling skin
effect in inductors, dielectric absorption in capacitors, and for converting white noise
into flicker noise.
An ideal fracpole exhibits the following impedance,
Z ( s ) = αs β .

(1)

which is shown in Figure 1.

FIGURE 1.

Impedance of an ideal fracpole.

Such a component is distributed if β is not an integer. It has time constants that are
spread over an infinite range, from infinitely long to infinitely short. In a transient-based
simulator, it must be approximated over a finite range of frequencies, from f0 to f1. The
component will be internally constructed as a lumped approximation that consists of a
parallel combination of resistors and capacitors, as shown in Figure 2. For this reason, β

FIGURE 2.

A lumped approximation to an ideal fracpole.

R0

R1

R2

...

RN

C1

C2

...

CN

Z(s)
C∞

is restricted to fall in the range – 1 < β < 0 . In this case, it represents a fractional impedance pole at 0 and is referred to as an impedance fracpole. If instead the approximation
is constructed with inductors replacing the capacitors, then β must fall in the range of
0 < β < 1. This represents a fractional admittance pole at 0 and is referred to as an
admittance fracpole. However, inductors are somewhat less efficiently modeled than
capacitors in most circuit simulators, and the same effect can be accomplished by combining a impedance fracpole with a gyrator.
An ideal fracpole is approximated over a finite range of frequencies by a circuit with
real pole and zero pairs such that the poles and zeros alternate and the pairs are evenly

spaced in a logarithmic sense over that range, as shown in Figure 3. The spacing

FIGURE 3.

Approximating an ideal fracpole using a collection of real poles and zeros.

between the pole and zero is the same for each pair, and this spacing relative to the spacing of the pairs themselves determines the slope, β, of the approximation. The number
of lumps, N, is chosen to provide a sufficient level of accuracy. The larger N, the closer
the impedance of the lumped approximation matches that of the ideal fracpole. Typically, using 1½ lumps per decade of frequency range give a good fit (but 1 lump per
decade is often sufficient).
The behavior outside the range of the approximation is determined by whether the first
and last critical frequencies are poles or zeros. Four possibilities exist, and they are distinguished by referring to them as different profiles. The profiles are named with a two
letter code, where the first letter refers to the low-frequency behavior and the second to
the high-frequency behavior. The letters are either d or f, where d signifies ‘down’ or a
down-slope and f signifies ‘flat’ or no slope. The profiles are shown in Figure 4.

FIGURE 4.

Log-impedance versus log-frequency for the various profiles.

Fractional Pole Modeler: FP

2.0 Fractional Pole Modeler: FP
The fp program takes a series of command line arguments and generates a Spectre subcircuit that models a fractional pole.
It takes the following arguments:
f0

The low frequency limit for the approximation (Hz).

f1

The high frequency limit for the approximation (Hz).

coef

The unity intercept point for the ideal impedance (the magnitude of the
impedance when ω=1 before approximation). The default value is 1.

slope

The slope of the impedance when plotted on a log-log scale (equals the negative of the fraction of a pole desired). Default is –0.5. Must be less than 0
and greater than –1.

lumps

The number of lumps used in the approximation (use lumps < 0 to specify
number of lumps/decade, in this case it need not be integer).

profile

Specifies whether the extreme critical frequencies of the impedance approximation are poles or zeros.

As an example, entering
fp f0=1 f1=1e6 slope=-0.5 lumps=-1.5 profile=dd
at the Unix command prompt generates the subcircuit given in Listing 1. It is placed in a
file named fracpole.scs and has the impedance, resistance, and reactance characteristics
shown in Figure 5.
Running the same command, except using lumps=–1 to get 1 lump (pole/zero pair) per
decade, gives the results shown in Figure 6. In this case there is noticeable ripple.
It is recommended that for best accuracy outside the approximation range, the DD profile is used when slope < –0.5 and the FF profile is used when slope > –0.5.

3.0 Inductor Modeler: IND
The ind program takes a series of command line arguments and generates a Spectre subcircuit that models an inductor that exhibits skin effect loss. The basic schematic of the
inductor is shown in Figure 7 [6]. The component H models skin effect and is implemented using a fracpole and a gyrator.
It takes the following arguments:

l

The inductance (H).

rs

The low frequency resistance (the ESR) (Ω).

cp

The shunt parasitic capacitance (F).

rp

The high frequency resistance (Ω).

h

The skin effect parameter.

LISTING 1.

FracPole subcircuit from the file “fracpole.scs”.
simulator lang=spectre
// Fractional Impedance Pole
// Impedance has a slope of –0.5 on a log-log scale,
// model is valid from 1 Hz to 1e+06 Hz.
subckt fracpole (p n)
parameters coef=1
R1 (p 1) resistor r=0.427978∗coef
C1 (1 n) capacitor c=0.253357/coef
R2 (p 2) resistor r=0.290855∗coef
C2 (2 n) capacitor c=0.0803174/coef
R3 (p 3) resistor r=0.143043∗coef
C3 (3 n) capacitor c=0.0351848/coef
R4 (p 4) resistor r=0.0671815∗coef
C4 (4 n) capacitor c=0.01614/coef
R5 (p 5) resistor r=0.0312686∗coef
C5 (5 n) capacitor c=0.007471/coef
R6 (p 6) resistor r=0.0145382∗coef
C6 (6 n) capacitor c=0.00346187/coef
R7 (p 7) resistor r=0.00678533∗coef
C7 (7 n) capacitor c=0.00159802/coef
R8 (p 8) resistor r=0.0032317∗coef
C8 (8 n) capacitor c=0.000722864/coef
R9 (p 9) resistor r=0.00172075∗coef
C9 (9 n) capacitor c=0.000292484/coef
Cinf (p n) capacitor c=0.000398944/coef
ends fracpole

FIGURE 5.

Impedance characteristics of fracpole with 1.5 lumps/decade.

f0

The low frequency limit for the skin effect approximation (Hz). It is computed automatically to be below the ESR corner frequency if not given.

f1

The high frequency limit for the skin effect model (Hz). It is computed automatically to be above the ESL resonant frequency if not given.

Inductor Modeler: IND

FIGURE 6.

Impedance characteristics of fracpole with 1 lump/decade.

lumps

The number of lumps used in the skin effect model (use lumps < 0 to specify
number of lumps/decade, in this case it need not be integer).

profile

Specifies whether the extreme critical frequencies of the skin effect approximation are poles or zeros.

name

The name that will be used for the subcircuit.

As an example, entering
ind name=a01t l=2.6e–9 rp=8 rs=0.001 cp=230e–15 h=704000 lumps=–1.5 \
profile=dd
at the Unix command prompt generates the subcircuit given in Listing 2. These are the
parameters associated with the Coilcraft a01t minispring inductor. The subcircuit name
was specified to be a01t, so the subcircuit is placed in a file named a01t.scs. The subcircuit has the resistance and reactance characteristics shown in Figure 8.
It is recommended that for best accuracy on a full-range model the DD profile be used.
It is not necessary to specify f0 or f1 for a full range model, but the size of the model can
be reduced by specifying a restricted range for the skin effect approximation using f0
and f1 and/or by reducing the number of lumps in the approximation.
It may be necessary to specify the scaling parameter on the subcircuit when using it to
make the R and C values reasonably sized, though the default value works well in most
cases.

LISTING 2.

Inductor subcircuit.
simulator lang=spectre
subckt a01t (1 2)
parameters scaling=1M gmin=1e-12
L (1 3) inductor l=2.6e-09
S (3 4) skin_effect s=1/(sqrt(2∗M_PI)∗704000)
Cp (1 5) capacitor c=2.3e-13
Rp (5 4) resistor r=8
Rs (4 2) resistor r=0.001
subckt skin_effect (1 2)
parameters s=1
G1 (1 2 3 0) gyrator r=sqrt(scaling)
A1 (3 0) fracpole coef=scaling/s
// Gyrator used to convert fractional impedance pole into a fractional admittance
pole
subckt gyrator (t1 b1 t2 b2)
parameters r=1kOhm
Gm1 (t1 b1 t2 b2) vccs gm=1/r
Gm2 (b2 t2 t1 b1) vccs gm=1/r
G1 (t1 b1) resistor r=1/gmin
G2 (t2 b2) resistor r=1/gmin
ends gyrator
// Fractional Impedance Pole
// Impedance has a slope of -0.5 on a log-log scale (valid from 25kHz to 65GHz.
subckt fracpole (p n)
parameters coef=1
R1 (p 1) resistor r=0.00278394∗coef
C1 (1 n) capacitor c=0.00157399/coef
R2 (p 2) resistor r=0.00196615∗coef
C2 (2 n) capacitor c=0.000508957/coef
R3 (p 3) resistor r=0.000999681∗coef
C3 (3 n) capacitor c=0.000228597/coef
R4 (p 4) resistor r=0.000484109∗coef
C4 (4 n) capacitor c=0.000107801/coef
R5 (p 5) resistor r=0.000232054∗coef
C5 (5 n) capacitor c=5.13586e-05/coef
R6 (p 6) resistor r=0.000111005∗coef
C6 (6 n) capacitor c=2.45185e-05/coef
R7 (p 7) resistor r=5.31306e-05∗coef
C7 (7 n) capacitor c=1.16984e-05/coef
R8 (p 8) resistor r=2.55506e-05∗coef
C8 (8 n) capacitor c=5.55525e-06/coef
R9 (p 9) resistor r=1.2561e-05∗coef
C9 (9 n) capacitor c=2.58058e-06/coef
R10 (p 10) resistor r=6.93017e-06∗coef
C10 (10 n) capacitor c=1.06815e-06/coef
Cinf (p n) capacitor c=1.56378e-06/coef
ends fracpole
ends skin_effect
ends a01t

Capacitor Modeler: DA

FIGURE 8.

Impedance characteristics of Coilcraft a01t minispring inductor.

The inductor modeler uses the fracpole program, which is constructs a fracpole as a parallel combination of series RC pairs and so β is restricted to be negative. Skin effect
requires that β be positive. To accommodate the limitations of the fracpole program, let
β = –½ and then use a gyrator to flip the sign. Choose
r2
α = --------------- ,
2πH

(2)

where r is the gyrator gain, which is chosen to provide proper scaling for the signal levels. Doing so results in the over all impedance for the combination of the gyrator and the
fracpole being
jf
Z ( f ) = ------- .
H

(3)

4.0 Capacitor Modeler: DA
The da program takes a series of command line arguments and generates a Spectre subcircuit that models a capacitor that exhibits dielectric absorption. The basic schematic of
the capacitor is shown in Figure 9 [7]. The component A is a fracpole that models
dielectric absorption.

FIGURE 9.

Capacitor model that includes dielectric absorption.

It takes the following arguments:
cinf

The high frequency limit for capacitance (F).

c0

The low frequency limit for capacitance (F).

tau0

The average time constant of dielectric dipoles (s).

alpha

Specifies the width of time constant distribution. Must be greater than 0 and
less than 1, with larger values implying a larger spread.

esr

The equivalent series resistance (Ω).

esl

The equivalent series inductance (H).

rleak

The leakage resistance (Ω).

f0

The low frequency limit for the dielectric absorption model (Hz).

f1

The high frequency limit for the dielectric absorption model (Hz).

lumps

The number of lumps used in the dielectric absorption approximation (use
lumps < 0 to specify number of lumps/decade, in this case it need not be
integer).

profile

Specifies whether the extreme critical frequencies of the skin effect approximation are poles or zeros.

name

The name that will be used for the subcircuit.

As an example, entering
da name=lossycap cinf=10e-9 c0=22.5e-9 tau0=1 alpha=0.75 f0=0.01 f1=1e6 \
esr=0.25 esl=8e-9 rleak=1e9 lumps=-1.5 profile=fd
at the Unix command prompt generates the subcircuit given in Listing 3. These are the
parameters of a 10 nF XR7 multilayer monolithic ceramic capacitor. The subcircuit
name was specified to be lossycap, so the subcircuit is placed in a file named lossycap.scs. The subcircuit has the resistance and reactance characteristics shown in
Figure 10.
For a full range model, specify f0 so that it is below the Rleak corner and specify f1 so
that it is above the ESL resonance. The the size of the model can be reduced by specifying a restricted frequency range for the dielectric approximation using f0 and f1 and/or
by reducing the number of lumps in the approximation.

5.0 Conclusion
The FracPole Suite is available from www.designers-guide.org/Modeling.
Eventually the fracpole component will be added to Spectre as a built-in primitive.
When that occurs, these utility programs will no longer be needed because the inductor
and capacitor models can be implemented as simple Spectre parameterized subcircuits.
At that point support for the utility programs fp, ind, and da will be retired.
5.1 If You Have Questions
If you have questions about what you have just read, feel free to post them on the Forum
section of The Designer’s Guide Community website. Do so by going to www.designersguide.org/Forum.

Postscript

LISTING 3.

Capacitor subcircuit.
simulator lang=spectre
//
// Capacitor model that include dielectric absorption
//
subckt lossycap (1 2)
L (1 3) inductor l=8e-09 r=0.25
C (3 2) capacitor c=1e-08
Rl (3 2) resistor r=1e+09
Cx (3 4) capacitor c=1.25e-08
DA (4 2) fracpole
// Fractional Impedance Pole
// Impedance has a slope of -0.75 on a log-log scale,
// model is valid from 0.01 Hz to 1e+06 Hz.
subckt fracpole (p n)
parameters coef=8e+07
R0 (p n) resistor r=7.96829∗coef
R1 (p 1) resistor r=9.42501∗coef
C1 (1 n) capacitor c=0.41576/coef
R2 (p 2) resistor r=2.40967∗coef
C2 (2 n) capacitor c=0.327729/coef
R3 (p 3) resistor r=0.703962∗coef
C3 (3 n) capacitor c=0.226085/coef
R4 (p 4) resistor r=0.210529∗coef
C4 (4 n) capacitor c=0.152355/coef
R5 (p 5) resistor r=0.0632518∗coef
C5 (5 n) capacitor c=0.102198/coef
R6 (p 6) resistor r=0.0190214∗coef
C6 (6 n) capacitor c=0.0684891/coef
R7 (p 7) resistor r=0.00572177∗coef
C7 (7 n) capacitor c=0.0458861/coef
R8 (p 8) resistor r=0.00172196∗coef
C8 (8 n) capacitor c=0.0307281/coef
R9 (p 9) resistor r=0.000519356∗coef
C9 (9 n) capacitor c=0.0205325/coef
R10 (p 10) resistor r=0.000158398∗coef
C10 (10 n) capacitor c=0.0135676/coef
R11 (p 11) resistor r=5.16126e-05∗coef
C11 (11 n) capacitor c=0.00839165/coef
Cinf (p n) capacitor c=0.0199735/coef
ends fracpole
ends lossycap

Postscript
I would like to thank Cairong (or Charlie) Hu for pointing out a flaw in the da program.
Prof. J. A. Tenreiro Machado has pointed out that the fracpole is an application of fractional calculus (FC) modeling, and that there is a well established community studying

FIGURE 10.

Impedance characteristics of 10 nF XR7 multilayer monolithic ceramic capacitor model.

FC and its applications. He continued by saying that the algorithm for modeling fractional poles with a collection of poles and zeros was first proposed by A. Oustaloup
some years ago [9] and that Bohannan also published some interesting work where
attempts to exploit the fractional order nature of capacitor, or what he calls ‘fractance’
[3], which is the characteristic exhibited by a ‘fractor’. He references work that presents
ideas similar to those presented in this paper, but based on a model of the dielectric
developed by Curie that dates back to 1889 [5,10].
Fracpoles also play a role in modeling batteries [2], where a fracpole is referred to as a
Warburg element and is used to model electrochemical diffusion. There are several diffusion models, but the simplest is the Warburg element [4]. A Warburg impedance element can be used to model semi-infinite linear diffusion, that is, unrestricted diffusion to
a large planar electrode. Its impedance is
AW
Z W = ---------.
jω

(4)

One models a parallel plate electrochemical cell using Randles’ circuit, which employs
a Warburg element to represent the electrochemical diffusion. Randles’ circuit is an
equivalent electrical circuit that consists of an active electrolyte resistance RS in series
with the parallel combination of the double-layer capacitance Cdl and the impedance of
a faradaic reaction. The impedance of a faradaic reaction consists of an active charge
transfer resistance Rct and a specific electrochemical element of diffusion AW, which is
the Warburg element. [1,8,11]

